Fuel Subsidy or Transfer?

Analysis of Small Closed Economy Using CGE Model

Prepared for Econ 296s Project

Justina Adamanti
Tevy Chawwa

DUKE UNIVERSITY

SPRING 2012

ABSTRACT

Sales tax imposed by government will raises the consumer prices and decreases the consumer purchasing power. This leads to a decline in household consumption and consequently decreases the utility of household. This paper uses CGE Model to analyze three alternatives compensation of government tax, first is by giving subsidy to a particular sector, second is by giving lump-sum transfer to household and third is by giving combination of subsidy and transfer. The result shows that utility of household is better when government compensate in terms of full transfer rather than combination of transfer and subsidy.

INTRODUCTION

Increase in the current world crude oil's price gives high pressure to several country governments' budget because the government subsidy the price of fuel in order to keep the domestic fuel price constant. In order to reduce the budget deficit, government can reduce the fuel subsidies. Although reduction of fuel subsidy would be positive for the government budget, it leads to arise in CPI inflation. A reduction in the government subsidy increases not only fuel prices but also prices of goods produced in other sectors through their input-output linkages with the oil and gas sector. As a consequence, demand and production for the outputs of different sectors decline as their prices increase. The decline in production, in turn, affects the incomes (and thus consumption) of household.

As compensation for the subsidy reduction, government can give money transfer to low income households. There are debates about the best policy that government should do to maintain welfare, keep giving fuel subsidy or give transfer to the household. This background motivates us to build a simple CGE model that could explain the impact of these alternative policies to the welfare.

The objective of this paper is to analyze the impact of government policies (tax, subsidy, transfers or combination) to the economy and analyze the best alternative policy to maximize welfare.

METHODOLOGY

The impact of sales tax, subsidy and transfer is evaluated with a Computable General Equilibrium (CGE) model using GAMS. The model in this paper is based on Autarky economic model (Gilbert and Tower, forthcoming book) with some extensions on government policies: sales tax, subsidy, transfers and government expenditure. The economy is simplified into 2 sectors/goods: fuel and non-fuel. Five scenarios will be analyzed:

1. Baseline: No sales tax, no subsidies and no transfer.
2. Case 1: Government imposes sales tax 1% for all goods but no subsidies and no transfer. All the government revenue is used for government expenditure.
3. Case 2: Government imposes sales tax 1% for all goods and gives subsidy 1% to fuel sector.
4. Case 3: Government imposes sales tax 1% for all goods and uses all the income for lump-sum transfer for household.
5. Case 4: Government imposes sales tax 1% for all goods and gives subsidy 1% to fuel sector and lump-sum transfer for household.

STRUCTURE OF MODEL

This small closed economy model has three economic agents, namely producer, household, and government. Figure 1 shows the full diagram of all economic agents and activities. The summary of the model features is as follow:

A. Production

- There are two sectors of production, fuel and non-fuel.
- There are two factors of production, labor and capital. The assumptions for factors of production are full employed and mobile across sectors.
- The firms attempt to maximize their profit given their production technology and fixed resources supply.
- Consumer price can be different from producer price, since the government can impose sales tax and subsidy in producer price.

- All of production output will be consumed by household and government.

B. Household

- Household choose their consumption bundle in order to maximize their utility subject to budget constraint/income.
- Household (which is also as producer) earn income from selling production output and transfer from government. This household income will be equal as income from wage and rent.
- Household consume production output.

C. Government

- The government earns revenue from sales tax and expends it for subsidy, transfer and government consumption.
- Government consumes production output.

Table 1 presents the notations of all the variables and parameters in this model. There are 12 variables and 7 parameters. Producer price of fuel (P_{Fuel}) is set as the *numeraire*,

Table 1 Variables and Parameters

Variables	Description	Parameters	Description
U	Utility level	α	Shift parameters in utility
P_i	Producer price	β_i	Share parameters in utility
PC_i	Consumer price (after tax and subsidy)	ρ_i	Elasticity of production
C_i	Consumption level	γ_i	Shift parameters in production
Q_i	Output level	δ_{ji}	Share parameters in production
R_j	Factor price	$taxr_i$	Rate of tax
F_{ji}	Factor use level	$subr_i$	Rate of subsidy
FS_j	Factor endowment		
Y	Household income		
GDP	Gross domestic product		
G_i	Government consumption		
TRF	Transfer to household		



Figure 1 Small Closed Economy with Tax, Subsidy and Transfer

Equations

This model consists of 10 equations as follow:

1. **Production function**

$$Q_i = \gamma_i\left(\delta_{ji}K_i{}^{\rho_i} + \left(1+\delta_{ji}\right)L_i{}^{\rho_i}\right)^{1/\rho_i}$$

The production function is CES with two inputs of production, labor and capital.

2. **Factor demand function**

$$R_j = P_i Q_i\left[\sum_j \delta_{ji}F_{ji}^{\rho_i}\right]^{-1}\delta_{ji}F_{ji}^{\rho_i-1}$$

The factor demand function is derived from the first order condition of profit maximization problem.

3. **Consumer price**

$$PC_i = P_i(1 + taxr_i - subr_i)$$

Consumer price can be different with producer price, since the government can imposed tax and subsidy to producer price.

4. **Household Income**

$$Y = \sum_i P_i.Q_i + TRF$$

Household earn income from selling production output and transfer from government.

5. **Demand function**

$$C_i = \frac{\beta_i.Y}{PC_i}$$

Household choose their consumption bundle in order to maximize their utility subject to budget constraint/income.

6. **Utility function**

$$U = \alpha\prod_i c_i^{\beta_i}$$

Utility function is based on the consumption and takes the Cobb-Douglas form with $0 < \beta < 1$.

7. Resource constraint

$$FS_j = \sum_i F_{ji}$$

The total supply of production factors have to be equal to the demand of production factors, since this model uses full employment assumption.

8. Quantity Balance

$$C_i = Q_i - G_i$$

The total output of production has to be equal to household and government consumption.

9. Government balance

$$\sum_i P_i Q_i \, taxr_i = \sum_i P_i Q_i \, subr_i + TRF + \sum_i PC_i G_i$$

The total government revenue has to be equal to total expenditure. Government earns revenue from sales tax and expends it for subsidy, transfer and consumption.

10. Gross domestic product

$$GDP = \sum_i PC_i Q_i + \sum_i PC_i G_i$$

Total GDP consists of total value of household consumption and total value of government consumption.

<u>Model closure for factor production (labor and capital)</u>

Factors of production in this model are fully employed and mobile across sectors. To implement this closure, supply of labor and capital is treated as exogenous variable, while price of factor and demand of factor are treated as endogenous variable.

SIMULATION RESULT

There are 4 cases, where all of the results are summarized in Table 2.

Case 1 : Tax 1%, No Subsidy, No Transfer

Sales tax increases the consumer price of both goods and reduce the amount of good that consumer can buy. The simulation shows that 1% sales tax will reduce the consumption of fuel by 6.14% and consumption of non-fuel goods by 11.71%. Consequently, household utility will be decline around 9%. Given the constant supply of production factors (capital and labor), the firms will make decision about quantity of production of each sector that can maximize their profit.

From profit maximization, the firms decide simultaneously the factor prices and optimal allocation of inputs, which can in turn be converted to quantity of output in both sectors. As a result, output of fuel will decreases by 6.14% and output of non-fuel increases by 5.95%. Demand of production factors (labor and capital) on fuel sector will decrease and demand of factors on non-fuel sector will increase. Wage rate will increase by 8.51% and rental rate will decrease by 2.01%.

In this model, the producer price of fuel is fixed as *numeraire,* and the increase in wage rate will lead to an increase in producer price of non-fuel to 1.06. As total production quantity and producer price increase, household income will increase by 3.25%. In order to exhaust the remaining output, government will expends all of the revenue on non-fuel goods. The result of case 1 is illustrated in Figure 2.

Table 2 Cost Benefit Analysis

Variable	Goods / Sector	Nota-tion	Baseline	Case 1		Case 2		Case 3		Case 4	
			No tax, No subsidy, No Transfer	Tax 1% , No subsidy, No Transfer	% Change	Tax 1%, Fuel Subsidy 1%, No Transfer	% Change	Tax 1%, No Subsidy, Transfer	% Change	Tax 1%, Subsidy 1%, Transfer	% Change
Utility		U	200	182.07	-8.96	190.69	-4.65	200	0	199.85	-0.08
Producer Price	Fuel	PF	1	1	0	1	0	1	0	1	0
	Non Fuel	PNF	1	1.06	6.31	1	0.00	1	0	0.97	-3.15
Consumer Price	Fuel	PCF	1	1.1	10.00	1	0.00	1.1	10.00	1	0
	Non Fuel	PCNF	1	1.17	16.94	1.1	10.00	1.1	10.00	1.07	6.53
Consumption	Fuel	CF	100	93.86	-6.14	100.00	0.00	100	0	103.14	3.14
	Non Fuel	CNF	100	88.29	-11.71	90.91	-9.09	100	0	96.81	-3.19
Output	Fuel	QF	100	93.86	-6.14	100.00	0.00	100	0	103.14	3.14
	Non Fuel	QNF	100	105.95	5.95	100.00	0.00	100	0	96.81	-3.19
Rent		RK	1	0.98	-2.01	1	0.00	1	0	1.01	1.08
Wage		RL	1	1.09	8.51	1	0.00	1	0	0.96	-4.18
Supply of Capital		FSK	100	100	0	100	0	100	0	100	0
Supply of Labor		FSL	100	100	0	100	0	100	0	100	0
Demand of Capital	Fuel	FKF	80	76.80	-4.00	80	0.00	80	0	81.54	1.92
	Non Fuel	FKNF	20	23.20	15.99	20	0.00	20	0	18.46	-7.68
Demand of Labor	Fuel	FLF	20	17.14	-14.28	20	0.00	20	0	21.63	8.14
	Non Fuel	FLNF	80	82.86	3.57	80	0.00	80	0	78.37	-2.04
GDP		GDP	200	227.15	13.57	210	5.00	220	10.00	206.28	3.14
Income		Y	200	206.50	3.25	200	0.00	220	10.00	206.28	3.14
Government Expenditure	Fuel	GF	0	0	0	0		0	0	0	0
	Non Fuel	GNF	0	17.66		9.09		0	0	0	0
Transfer		TRF	0	0	0	0	0	20		9.38	0



Figure 2 Illustration of Case 1: Impact of Sales Tax

Case 2 : Tax 1%, Fuel Subsidy 1%, No Transfer

In this case, government uses some of their revenue to subsidy fuel price; therefore the consumer price of fuel will be the same with producer price. Thus, consumer will consume the same amount of fuel as before the tax imposed and will consume less non fuel goods. Compared to the baseline, utility of household will decrease about 4.65%. Quantity of production in both sector will remain constant, and so does the demand and price of factor productions. Government will expend the remaining revenue on non-fuel goods. The result of case 1 is summarized in Figure 3 below.



Figure 3 Illustration of Case 2: Impact of Sales Tax and Fuel Subsidy

Case 3 : Tax 1%, Transfer, No Subsidy

In this case, government will transfers all of their revenue to the household. Thus, although the prices of both goods are higher because of sales tax, it is still affordable for household to buy the same amount of those goods because now they have additional income. The utility of household will be the same as before the tax imposed. There will be no impact on production sector. The illustration of this case can be seen by combining production possibility curve and consumer indifference curve in Figure 4. The budget constraint of after tax plus transfer will be the same with budget constraint before tax (line AB), and so does the indifference curve and utility (U_3).

Case 4: Tax 1%, Fuel Subsidy 1% and Transfer

In this case, government uses part of their revenue to subsidize the price of fuel sector 1% and uses the remaining for transfer to household. The fuel subsidy will make the relative consumer price of fuel (P_F/P_{NF}) less than 1, and move the budget line from AB to A'B' (Figure 5). Then, transfers will shifts up the budget line from A'B' to A''B''. The new equilibrium point shows that consumer will consume more fuel and less non-fuel, and producer also produce more fuel and less non-fuel. However, the indifference curve of the new equilibrium point shows that the utility of household is lower than the utility gained in case 3. The simulation result shows that the utility in this case is 0.08% lower. Since the production of fuel increases, there will be movement of production factors from non-fuel sector to fuel sector and the rental rate will increase. In the opposite, production of non-fuel decreases and the wage will decreases since this sector is labor intensive.



Figure 4 Production Possibility Frontier and Indifference Curve in Case 3



Figure 5 Comparison between Production Possibility Frontier and Indifference Curve in Case 3 and Case 4

CONCLUSION

The imposition of sales tax will reduce the welfare. The simulation shows that government can maintain the welfare better by give money transfers to household as compensation rather than subsidy.

REFERENCE

Gilbert, J., Tower, E. *"An Introduction to GAMS Modeling for International Trade Theory and Policy"* (forthcoming).

Friedman, M. "*The Welfare Effects of an Income Tax and an Excise Tax*". *Journal of Political Economy* Vol. 60, No. 1 (Feb., 1952), pp. 25-33

APPENDIX

```
*Small Closed Economy with Tax, Subsidy and Transfer
*Based on Autarky Model (Gilbert & Tower)

*Econ 296 - Project 2 Spring 2012
*Tevy Chawwa & Justina Adamanti

SETS
I Goods /FUEL, NFUEL/
J Factors /K,L/
;
ALIAS (J, JJ), (I,II);

PARAMETERS
ALPHA            Shift parameters in utility
BETA(I)          Share parameters in utility
GAMMA(I)         Shift parameters in production
DELTA(J,I)       Share parameters in production
RHO(I)           Elasticity parameters in production
TAXR(I)          Tax Rate
SUBR(I)          Subsidy Rate
PO(I)            Initial prices
PCO(I)           After Tax Price
UO               Initial utility level
GO(I)            Government Expenditure
CO(I)            Initial consumption levels
QO(I)            Initial output levels
FSO(J)           Initial Endowments
FO(J,I)          Initial factor use levels
RO(J)            Initial factor prices
GDPO             Initial gross domestic product
YO               Household or Production Income
TRFO             Total Transfer
;

*INITIALIZATION & CALIBRATION-----------------------------------------
*prices (good & factor)
PO(I)= 1;
RO(J)= 1;

TAXR(I) = 0;
SUBR(I) = 0;
PCO(I)  = PO(I)*(1+TAXR(I)-SUBR(I));

*Production & Consumption
QO(I)= 100;
GO(I)= 0;
CO(I)= QO(I)-GO(I);

*Supply and demand of Factor
FO('L','FUEL')   = 20;
FO('L','NFUEL')  = 80;
FO('K',I)        = (QO(I)*PO(I)-FO('L',I)*RO('L'))/RO('K');
FSO(J)           = SUM(I, FO(J,I));

*CES Production Function
RHO(I)    = 0.1;
DELTA(J,I)= (RO(J)/FO(J,I)**(RHO(I)-1))/(SUM(JJ, RO(JJ)/FO(JJ,I)**(RHO(I)-1)));
GAMMA(I)  = QO(I)/(SUM(J, DELTA(J,I)*FO(J,I)**RHO(I)))**(1/RHO(I));
```

```
*Income
TRFO = 0;
YO   = SUM(I, PO(I)*QO(I))+TRFO;
BETA(I)  = CO(I)*PCO(I)/YO;

*Utility
UO       = YO;
ALPHA    = UO/PROD(I, CO(I)**BETA(I));

*GDP
GDPO     = SUM(I, PCO(I)*CO(I))+ SUM(I,PCO(I)*GO(I));

**-----------------------------------------END OF INITIALIZATION & CALIBRATION---

DISPLAY PO, RO, QO, CO, FO, GO, GDPO, RHO, DELTA, GAMMA, UO, BETA, ALPHA;

VARIABLES
U        Utility level
P(I)     Prices
C(I)     Consumption levels
Q(I)     Output levels
R(J)     Factor prices
F(J,I)   Factor use levels
FS(J)    Endowments
GDP      Gross domestic product
Y        Income
PC(I)    Sales Price
G(I)     Government Expenditure
TRF      Transfer
;

U.L      = UO;
P.L(I)   = PO(I);
C.L(I)   = CO(I);
Q.L(I)   = QO(I);
R.L(J)   = RO(J);
F.L(J,I) = FO(J,I);
FS.L(J)  = FSO(J);
GDP.L    = GDPO;
Y.L      = YO;
PC.L(I)  = PCO(I);
G.L (I)  = GO (I);
TRF.L    = TRFO;

P.LO(I)  = 0;
C.LO(I)  = 0;
Q.LO(I)  = 0;
R.LO(J)  = 0;
F.LO(J,I)= 0;
FS.LO(J) = 0;
GDP.LO   = 0;
Y.LO     = 0;
PC.LO(I) = 0;
G.LO(I)  = 0;
TRF.LO   = 0;

**FACTOR MARKET CLOSURE-------------------------------------------------------
scalar FMCLOS Closure for Factor Market /1/;

if (FMCLOS eq 1,
*Fully employed  & Mobile
  R.LO(J) = 0;
  R.UP(J) = +INF;
```

```
  R.L(J)  = RO(J);

  FS.FX(J) = FSO(J);

  F.LO(J,I) = 0;
  F.UP(J,I) = +INF;
  F.L(J,I)  = FO(J,I);
);

IF (FMCLOS eq 2,
*Fully employed & Not Mobile
  R.FX(J) = RO(J);

  FS.LO(J) = 0;
  FS.UP(J) = +INF;
  FS.L(J)  = FSO(J);

  F.FX(J,I)  = FO(J,I);
);
**-------------------------------------------------END OF FACTOR MARKET CLOSURE---

*NUMERAIRE---------------------------------------------------------------------
P.FX('FUEL')= 1;
*------------------------------------------------------------------------------

EQUATIONS
UTILITY          Utility function

PRODUCTION(I)    Production functions
FDEMAND(J,I)     Factor demand functions
CPRICE(I)        Consumen Price after tax

INCOME           Household or Production Income
EQ_GDP           Gross Domestic Product
DEMAND(I)        Demand functions

RESOURCE(J)      Resource constraints
MAT_BAL(I)       Market clearing
GOVBAL           Government Balance
;

UTILITY..        U =E= ALPHA*PROD(I, C(I)**BETA(I));

PRODUCTION(I)..  Q(I) =E= GAMMA(I)*SUM(J, DELTA(J,I)*F(J,I)**RHO(I))**(1/RHO(I));
FDEMAND(J,I)..   R(J) =E= P(I)*Q(I)*SUM(JJ, DELTA(JJ,I)*F(JJ,I)**RHO(I))
**(-1)*DELTA(J,I)*F(J,I)**(RHO(I)-1);
CPRICE(I)..      PC(I) =E= P(I)*(1+TAXR(I)-SUBR(I));

INCOME..         Y =E= SUM(I, P(I)*Q(I))+TRF;
DEMAND(I)..      C(I) =E= BETA(I)*Y/PC(I);
EQ_GDP..         GDP =E= SUM(I, PC(I)*C(I))+SUM(I,G(I)*PC(I));

RESOURCE(J)..    FS(J) =E= SUM(I, F(J,I));
MAT_BAL(I)..     C(I) =E= Q(I)-G(I);
GOVBAL..         SUM(I, P(I)*Q(I)*TAXR(I)) =E= SUM(I,
P(I)*Q(I)*SUBR(I))+TRF+sum(I,G(I)*PC(I));

*SOLVE BASE MODEL--------------------------------------------------------------
MODEL AUTARKY /ALL/;
SOLVE AUTARKY USING NLP MAXIMIZING U;

*REPORT------------------------------------------------------------------------
SET
```

```
SIM simulations
/BASE base simulation
 SIMN increase
 C changes in percent/;

parameters
 U_R(SIM), P_R(I,SIM), C_R(I,SIM), Q_R(I,SIM), R_R(J,SIM)
 FS_R(J, SIM), F_R(J,I,SIM), GDP_R(SIM), G_R(I,SIM)
 PC_R(I, SIM), TRF_R(SIM), Y_R(SIM)
;

U_R('BASE')       = U.L;
P_R(I,'BASE')     = P.L(I);
C_R(I,'BASE')     = C.L(I);
Q_R(I,'BASE')     = Q.L(I);
R_R(J,'BASE')     = R.L(J);
FS_R(J,'BASE')    = FS.L(J);
F_R(J,I,'BASE')   = F.L(J,I);
Y_R('BASE')       = Y.L;
GDP_R('BASE')     = GDP.L;
PC_R(I,'BASE')    = PC.L(I);
G_R(I,'BASE')     = G.L(I);
TRF_R('BASE')       = TRF.L;

**SIMULATION-------------------------------------------------------------------------
--
P.FX('FUEL')= 1;

*G(I).FX = 0;
SUBR('FUEL') = 0;
TAXR(I)= 0.1;
TRF.FX = 0;

*------------------------------------------------------------------------------------
MODEL AUTARKY_SIM /ALL/;
SOLVE AUTARKY_SIM USING NLP MAXIMIZING U;
**------------------------------------------------------------END OF SIMULATION--
*REPORT----------------------------------------------------------------------------
U_R('SIMN')       = U.L;
P_R(I,'SIMN')     = P.L(I);
C_R(I,'SIMN')     = C.L(I);
Q_R(I,'SIMN')     = Q.L(I);
R_R(J,'SIMN')     = R.L(J);
FS_R(J,'SIMN')    = FS.L(J);
F_R(J,I,'SIMN')   = F.L(J,I);
GDP_R('SIMN')     = GDP.L;
Y_R('SIMN')       = Y.L;
PC_R(I,'SIMN')    = PC.L(I);
G_R(I,'SIMN')       = G.L(I);
TRF_R('SIMN')      = TRF.L;

*%changes
U_R('C')       = 100*((U_R('SIMN')/U_R('BASE'))-1);
P_R(I,'C')     = 100*((P_R(I,'SIMN')/P_R(I,'BASE'))-1);
C_R(I,'C')     = 100*((C_R(I,'SIMN')/C_R(I,'BASE'))-1);
Q_R(I,'C')     = 100*((Q_R(I,'SIMN')/Q_R(I,'BASE'))-1);
R_R(J,'C')     = 100*((R_R(J,'SIMN')/R_R(J,'BASE'))-1);
FS_R(J,'C')    = 100*((FS_R(J,'SIMN')/FS_R(J,'BASE'))-1);
F_R(J,I,'C')   = 100*((F_R(J,I,'SIMN')/F_R(J,I,'BASE'))-1);
GDP_R('C')     = 100*((GDP_R('SIMN')/GDP_R('BASE'))-1);
Y_R('C')       = 100*((Y_R('SIMN')/Y_R('BASE'))-1);
PC_R(I,'C')    = 100*((PC_R(I,'SIMN')/PC_R(I,'BASE'))-1);
*G_R(I,'C')     = 100*((G_R(I,'SIMN')/G_R(I,'BASE'))-1);
```

```
*TRF_R('C')       = 100*((TRF_R('SIMN')/TRF_R('BASE'))-1);
**------------------------------------------------------------ END OF REPORT---

display
U_R, P_R, PC_R, C_R, Q_R, R_R, FS_R, F_R, GDP_R, Y_R, G_R, TRF_R;
```